

NAG Library Routine Document

G05THF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05THF generates a vector of pseudorandom integers from the discrete negative binomial distribution with parameter m and probability p of success at a trial.

2 Specification

```
SUBROUTINE G05THF(MODE, N, M, P, R, LR, STATE, X, IFAIL)
INTEGER          MODE, N, M, LR, STATE(lstate), X(N), IFAIL
double precision P, R(LR)
```

3 Description

G05THF generates n integers x_i from a discrete negative binomial distribution, where the probability of $x_i = I$ (I successes before m failures) is

$$P(x_i = I) = \frac{(m + I - 1)!}{I!(m - 1)!} \times p^I \times (1 - p)^m, \quad I = 0, 1, \dots$$

The variates can be generated with or without using a search table and index. If a search table is used then it is stored with the index in a reference vector and subsequent calls to G05THF with the same parameter value can then use this reference vector to generate further variates.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05THF.

4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison–Wesley

5 Parameters

1: MODE – INTEGER *Input*

On entry: a code for selecting the operation to be performed by the routine.

MODE = 0

Set up reference vector only.

MODE = 1

Generate variates using reference vector set up in a prior call to G05THF.

MODE = 2

Set up reference vector and generate variates.

MODE = 3

Generate variates without using the reference vector.

Constraint: MODE = 0, 1, 2 or 3.

- 2: N – INTEGER *Input*
On entry: n , the number of pseudorandom numbers to be generated.
Constraint: $N \geq 0$.
- 3: M – INTEGER *Input*
On entry: m , the number of failures of the distribution.
Constraint: $M \geq 0$.
- 4: P – **double precision** *Input*
On entry: p , the parameter of the negative binomial distribution representing the probability of success at a single trial.
Constraint: $0.0 \leq P < 1.0$.
- 5: R(LR) – **double precision** array *Communication Array*
On entry: if $\text{MODE} = 1$, the reference vector from the previous call to G05THF.
 If $\text{MODE} = 3$, R is not referenced by G05THF.
On exit: the reference vector.
- 6: LR – INTEGER *Input*
On entry: the dimension of the array R as declared in the (sub)program from which G05THF is called.
Suggested value:
 if $\text{MODE} \neq 3$,
 $\text{LR} = 28 + \left(20 \times \sqrt{M \times P} + 30 \times P \right) / (1 - P)$ approximately;
 otherwise $\text{LR} = 1$.
Constraints:
 if $\text{MODE} = 0$ or 2 ,
 $\text{LR} > \text{int} \left(\frac{M \times P + 7.15 \times \sqrt{M \times P} + 20.15 \times P}{1 - P} + 8.5 \right)$
 $- \max \left(0, \text{int} \left(\frac{M \times P - 7.15 \times \sqrt{M \times P}}{1 - P} \right) \right) + 9$;
 if $\text{MODE} = 1$, LR must remain unchanged from the previous call to G05THF.
- 7: STATE(*lstate*) – INTEGER array *Communication Array*
Note: *lstate* = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.
On entry: contains information on the selected base generator and its current state.
On exit: contains updated information on the state of the generator.
- 8: X(N) – INTEGER array *Output*
On exit: the n pseudorandom numbers from the specified negative binomial distribution.
- 9: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $MODE \neq 0, 1, 2$ or 3 .

$IFAIL = 2$

On entry, $N < 0$.

$IFAIL = 3$

On entry, $M < 0$.

$IFAIL = 4$

On entry, $P < 0.0$,
or $P \geq 1.0$.

$IFAIL = 5$

On entry, P or M is not the same as when R was set up in a previous call to G05THF with $MODE = 0$ or 2 .

On entry, the R vector was not initialized correctly, or has been corrupted.

$IFAIL = 6$

On entry, LR is too small when $MODE = 0$ or 2 .

$IFAIL = 7$

On entry, $STATE$ vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example prints 20 pseudorandom integers from a negative binomial distribution with parameters $m = 60$ and $p = 0.999$, generated by a single call to G05THF, after initialization by G05KFF.

9.1 Program Text

```

*      G05THF Example Program Text
*      Mark 22 Release. NAG Copyright 2006.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSEED, MSTATE, N, LR
      PARAMETER        (MSEED=1,MSTATE=633,N=20,LR=1)
*      .. Local Scalars ..
      DOUBLE PRECISION P
      INTEGER          GENID, I, IFAIL, LSEED, LSTATE, M, SUBID
*      .. Local Arrays ..
      DOUBLE PRECISION R(LR)
      INTEGER          SEED(MSEED), STATE(MSTATE), X(N)
*      .. External Subroutines ..
      EXTERNAL         G05KFF, G05THF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G05THF Example Program Results'
      WRITE (NOUT,*)
*      Set the distribution parameter P
      P = 0.999D0
      M = 60
*      Initialize the seed
      SEED(1) = 1762543
*      GENID and SUBID identify the base generator
      GENID = 1
      SUBID = 1
*      Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 20
      END IF
*      Generate the variates
*      Choose MODE = 3 because P close to 1
      IFAIL = 1
      CALL G05THF(3,N,M,P,R,LR,STATE,X,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 20
      END IF
*      Display the variates
      WRITE (NOUT,99999) (X(I),I=1,N)
*
      20 CONTINUE
*
99999  FORMAT (1X,I12)
99998  FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997  FORMAT (1X,' ** G05THF returned with IFAIL = ',I5)
      END

```

9.2 Program Data

None.

9.3 Program Results

G05THF Example Program Results

```

62339
50505
64863
66289
50434
59461
57365

```

65965
59572
63104
47833
54735
62075
48018
61458
55190
54263
80995
70129
60200
